

```

1 ;-----
2 ;
3 ; The WOZ Monitor for the Apple 1
4 ; Written by Steve Wozniak 1976
5 ;
6 ;-----
7
8         .CR      6502
9         .OR      $FF00
10        .TF      WOZMON.HEX, HEX, 8
11
12 ;-----
13 ; Memory declaration
14 ;-----
15
16 XAML      .EQ      $24      Last "opened" location Low
17 XAMH      .EQ      $25      Last "opened" location High
18 STL       .EQ      $26      Store address Low
19 STH       .EQ      $27      Store address High
20 L         .EQ      $28      Hex value parsing Low
21 H         .EQ      $29      Hex value parsing High
22 YSAV      .EQ      $2A      Used to see if hex value is given
23 MODE      .EQ      $2B      $00=XAM, $7F=STOR, $AE=BLOCK XAM
24
25 IN        .EQ      $0200,$027F      Input buffer
26
27 KBD       .EQ      $D010      PIA.A keyboard input
28 KBDCCR    .EQ      $D011      PIA.A keyboard control register
29 DSP       .EQ      $D012      PIA.B display output register
30 DSPCR     .EQ      $D013      PIA.B display control register
31
32 ; KBD b7..b0 are inputs, b6..b0 is ASCII input, b7 is constant high
33 ;   Programmed to respond to low to high KBD strobe
34 ; DSP b6..b0 are outputs, b7 is input
35 ;   CB2 goes low when data is written, returns high when CB1 goes high
36 ; Interrupts are enabled, though not used. KBD can be jumpered to IRQ,
37 ; whereas DSP can be jumpered to NMI.
38
39 ;-----
40 ; Constants
41 ;-----
42
43 BS        .EQ      $DF      Backspace key, arrow left key
44 CR        .EQ      $8D      Carriage Return
45 ESC       .EQ      $9B      ESC key
46 PROMPT    .EQ      "\"      Prompt character
47
48 ;-----
49 ; Let's get started
50 ;
51 ; Remark the RESET routine is only to be entered by asserting the RESET
52 ; line of the system. This ensures that the data direction registers
53 ; are selected.
54 ;-----
55
56 RESET     CLD              Clear decimal arithmetic mode
57           CLI
58           LDY      #0111.1111      Mask for DSP data direction reg
59           STY      DSP              (DDR mode is assumed after reset)
60           LDA      #1010.0111      KBD and DSP control register mask
61           STA      KBDCCR           Enable interrupts, set CA1, CB1 for
62           STA      DSPCR            positive edge sense/output mode.
63
64 ; Program falls through to the GETLINE routine to save some program bytes
65 ; Please note that Y still holds $7F, which will cause an automatic Escape
66
67 ;-----
68 ; The GETLINE process
69 ;-----

```

70				
71	NOTCR	CMP	#BS	Backspace key?
72		BEQ	BACKSPACE	Yes
73		CMP	#ESC	ESC?
74		BEQ	ESCAPE	Yes
75		INY		Advance text index
76		BPL	NEXTCHAR	Auto ESC if line longer than 127
77				
78	ESCAPE	LDA	#PROMPT	Print prompt character
79		JSR	ECHO	Output it.
80				
81	GETLINE	LDA	#CR	Send CR
82		JSR	ECHO	
83				
84		LDY	#0+1	Start a new input line
85	BACKSPACE	DEY		Backup text index
86		BMI	GETLINE	Oops, line's empty, reinitialize
87				
88	NEXTCHAR	LDA	KBD CR	Wait for key press
89		BPL	NEXTCHAR	No key yet!
90		LDA	KBD	Load character. B7 should be '1'
91		STA	IN,Y	Add to text buffer
92		JSR	ECHO	Display character
93		CMP	#CR	
94		BNE	NOTCR	It's not CR!
95				
96	; Line received, now let's parse it			
97				
98		LDY	#-1	Reset text index
99		LDA	#0	Default mode is XAM
100		TAX		X=0
101				
102	SETSTOR	ASL		Leaves \$7B if setting STOR mode
103				
104	SETMODE	STA	MODE	Set mode flags
105				
106	BLSKIP	INY		Advance text index
107				
108	NEXTITEM	LDA	IN,Y	Get character
109		CMP	#CR	
110		BEQ	GETLINE	We're done if it's CR!
111		CMP	#"."	
112		BCC	BLSKIP	Ignore everything below "."!
113		BEQ	SETMODE	Set BLOCK XAM mode ( "." = \$AE)
114		CMP	#":"	
115		BEQ	SETSTOR	Set STOR mode! \$BA will become \$7B
116		CMP	#"R"	
117		BEQ	RUN	Run the program! Forget the rest
118		STX	L	Clear input value (X=0)
119		STX	H	
120		STY	YSAV	Save Y for comparison
121				
122	; Here we're trying to parse a new hex value			
123				
124	NEXTHEX	LDA	IN,Y	Get character for hex test
125		EOR	#\$B0	Map digits to 0-9
126		CMP	#9+1	Is it a decimal digit?
127		BCC	DIG	Yes!
128		ADC	#\$88	Map letter "A"-"F" to \$FA-FF
129		CMP	#\$FA	Hex letter?
130		BCC	NOTHEX	No! Character not hex
131				
132	DIG	ASL		
133		ASL		Hex digit to MSD of A
134		ASL		
135		ASL		
136				
137		LDX	#4	Shift count
138	HEXSHIFT	ASL		Hex digit left, MSB to carry

```

139             ROL     L             Rotate into LSD
140             ROL     H             Rotate into MSD's
141             DEX                     Done 4 shifts?
142             BNE     HEXSHIFT      No, loop
143             INY                     Advance text index
144             BNE     NEXTHEX       Always taken
145
146 NOTHEX             CPY     YSAV     Was at least 1 hex digit given?
147             BEQ     ESCAPE        No! Ignore all, start from scratch
148
149             BIT     MODE          Test MODE byte
150             BVC     NOTSTOR       B6=0 is STOR, 1 is XAM or BLOCK XAM
151
152 ; STOR mode, save LSD of new hex byte
153
154             LDA     L             LSD's of hex data
155             STA     (STL,X)       Store current 'store index' (X=0)
156             INC     STL           Increment store index.
157             BNE     NEXTTITEM     No carry!
158             INC     STH           Add carry to 'store index' high
159 TONEXTTITEM        JMP     NEXTTITEM Get next command item.
160
161 ;-----
162 ; RUN user's program from last opened location
163 ;-----
164
165 RUN             JMP     (XAML)     Run user's program
166
167 ;-----
168 ; We're not in Store mode
169 ;-----
170
171 NOTSTOR         BMI     XAMNEXT    B7 = 0 for XAM, 1 for BLOCK XAM
172
173 ; We're in XAM mode now
174
175             LDX     #2            Copy 2 bytes
176 SETADR          LDA     L-1,X      Copy hex data to
177             STA     STL-1,X       'store index'
178             STA     XAML-1,X      and to 'XAM index'
179             DEX                     Next of 2 bytes
180             BNE     SETADR        Loop unless X = 0
181
182 ; Print address and data from this address, fall through next BNE.
183
184 NXTPRNT         BNE     PRDATA     NE means no address to print
185             LDA     #CR           Print CR first
186             JSR     ECHO
187             LDA     XAMH          Output high-order byte of address
188             JSR     PRBYTE
189             LDA     XAML          Output low-order byte of address
190             JSR     PRBYTE
191             LDA     #":"          Print colon
192             JSR     ECHO
193
194 PRDATA          LDA     #" "      Print space
195             JSR     ECHO
196             LDA     (XAML,X)      Get data from address (X=0)
197             JSR     PRBYTE        Output it in hex format
198 XAMNEXT         STX     MODE       0 -> MODE (XAM mode).
199             LDA     XAML          See if there's more to print
200             CMP     L
201             LDA     XAMH
202             SBC     H
203             BCS     TONEXTTITEM   Not less! No more data to output
204
205             INC     XAML          Increment 'examine index'
206             BNE     MOD8CHK       No carry!
207             INC     XAMH

```

```

208
209 MOD8CHK          LDA      XAML          If address MOD 8 = 0 start new line
210                  AND      #%0000.0111
211                  BPL      NXTPRNT       Always taken.
212
213 ;-----
214 ; Subroutine to print a byte in A in hex form (destructive)
215 ;-----
216
217 PRBYTE            PHA              Save A for LSD
218                  LSR
219                  LSR
220                  LSR              MSD to LSD position
221                  LSR
222                  JSR      PRHEX      Output hex digit
223                  PLA              Restore A
224
225 ; Fall through to print hex routine
226
227 ;-----
228 ; Subroutine to print a hexadecimal digit
229 ;-----
230
231 PRHEX             AND      #%0000.1111   Mask LSD for hex print
232                  ORA      #"0"          Add "0"
233                  CMP      #"9"+1       Is it a decimal digit?
234                  BCC      ECHO          Yes! output it
235                  ADC      #6           Add offset for letter A-F
236
237 ; Fall through to print routine
238
239 ;-----
240 ; Subroutine to print a character to the terminal
241 ;-----
242
243 ECHO              BIT      DSP          DA bit (B7) cleared yet?
244                  BMI      ECHO          No! Wait for display ready
245                  STA      DSP          Output character. Sets DA
246                  RTS
247
248 ;-----
249 ; Vector area
250 ;-----
251
252                  .DA      $0000         Unused, what a pity
253 NMI_VEC           .DA      $0F00         NMI vector
254 RESET_VEC         .DA      RESET       RESET vector
255 IRQ_VEC           .DA      $0000         IRQ vector
256
257 ;-----
258
259                  .LI      OFF
260

```